General Linear Models for Testing Moderation: Multiple-Slope Interactions

- Topics:
  - Review of main effects of categorical predictors
    - Specification using dummy codes
    - Specification as “categorical” directly in the model syntax
  - Review of interaction concepts
  - Examples of interactions that require multiple slopes
    - Interactions among categorical predictors
    - Interactions with quantitative predictors with nonlinear effects
  - Special uses of interaction terms to create nested effects
    - “ANOVA with a hole in it”
    - Missing (or impossible) predictor data

Categorical Predictors (3+ Groups)

- Two alternatives for how to include grouping predictors

1. **Manually create** and include dummy-coded group contrasts
   - Need $C-1$ contrasts for C categories, added all at once, **treated as quantitative** (WITH in SPSS, by default in SAS, c. in STATA)
   - Corresponds more directly to linear model representation
   - Can be easier to set own reference group and contrasts of interest

2. **Let the program** create and include group contrasts for you
   - **Treated as categorical**: BY in SPSS, CLASS in SAS, i. in STATA
     - SPSS and SAS: reference = highest/last group; STATA: reference = lowest/first group
   - Can be more convenient if you have many groups, want many contrasts, or have interactions among grouping predictors
   - But it marginalizes over main effects when estimating other effects ☹

Categorical Predictors Via Manual Contrasts

- Model: $y_i = \beta_0 + \beta_1 d1_i + \beta_2 d2_i + \beta_3 d3_i + e_i$
  - "group" variable: Control=0, Treat1=1, Treat2=2, Treat3=3
  - New variables to be created for the model:
    d1= 0, 1, 0, 0 → difference between Control and T1
    d2= 0, 0, 1, 0 → difference between Control and T2
    d3= 0, 0, 0, 1 → difference between Control and T3
- How does the model give us **all possible group differences**? By determining each group's mean, and then the difference…

Control Mean (Reference)	Treatment 1 Mean	Treatment 2 Mean	Treatment 3 Mean
β_0	$\beta_0 + \beta_1 d1_i$	$\beta_0 + \beta_2 d2_i$	$\beta_0 + \beta_3 d3_i$

- The model for the 4 groups directly provides 3 differences (control vs. each treatment), and indirectly provides another 3 differences (differences between treatments)

Categorical Predictors Via Manual Contrasts

- Model: $y_i = \beta_0 + \beta_1 d1_i + \beta_2 d2_i + \beta_3 d3_i + e_i$

Control Mean (Reference)	Treatment 1 Mean	Treatment 2 Mean	Treatment 3 Mean
$\boldsymbol{\beta_0}$	$\boldsymbol{\beta_0 + \beta_1 d1_i}$	$\boldsymbol{\beta_0 + \beta_2 d2_i}$	$\boldsymbol{\beta_0 + \beta_3 d3_i}$

	Alt Group		Ref Group	Difference
• Control vs. T1 =	$(\beta_0+\beta_1)$	$-$	(β_0)	$= \beta_1$
• Control vs. T2 =	$(\beta_0+\beta_2)$	$-$	(β_0)	$= \beta_2$
• Control vs. T3 =	$(\beta_0+\beta_3)$	$-$	(β_0)	$= \beta_3$
• T1 vs. T2 =	$(\beta_0+\beta_2)$	$-$	$(\beta_0+\beta_1)$	$= \beta_2 - \beta_1$
• T1 vs. T3 =	$(\beta_0+\beta_3)$	$-$	$(\beta_0+\beta_1)$	$= \beta_3 - \beta_1$
• T2 vs. T3 =	$(\beta_0+\beta_3)$	$-$	$(\beta_0+\beta_2)$	$= \beta_3 - \beta_2$

Main Effects via Manual Contrasts: SAS

	Alt Group	Ref Group	Difference
• Control vs. T1 =	$(\beta_0+\beta_1)$	$-\ (\beta_0)$	$= \beta_1$
• Control vs. T2 =	$(\beta_0+\beta_2)$	$-\ (\beta_0)$	$= \beta_2$
• Control vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0)$	$= \beta_3$
• T1 vs. T2 =	$(\beta_0+\beta_2)$	$-\ (\beta_0+\beta_1)$	$= \beta_2 - \beta_1$
• T1 vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0+\beta_1)$	$= \beta_3 - \beta_1$
• T2 vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0+\beta_2)$	$= \beta_3 - \beta_2$

Note the order of the equations: the reference group mean *is subtracted from* the alternative group mean.

In SAS ESTIMATE statements (or SPSS TEST or STATA LINCOM), the variables refer to their betas; the numbers refer to the operations of their betas.

```
TITLE "SAS Manual Contrasts for 4-Group Diffs";
PROC GLM DATA=work.dataname NAMELEN=100;
MODEL y = d1 d2 d3 / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
CONTRAST "Omnibus DF=3 main effect F-test" d1 1, d2 1, d3 1;
ESTIMATE "Control Mean" intercept 1 d1 0 d2 0 d3 0;
ESTIMATE "T1 Mean"      intercept 1 d1 1 d2 0 d3 0;
ESTIMATE "T2 Mean"      intercept 1 d1 0 d2 1 d3 0;
ESTIMATE "T3 Mean"      intercept 1 d1 0 d2 0 d3 1;

ESTIMATE "Control vs. T1"        d1  1 d2  0 d3 0;
ESTIMATE "Control vs. T2"        d1  0 d2  1 d3 0;
ESTIMATE "Control vs. T3"        d1  0 d2  0 d3 1;

ESTIMATE "T1 vs. T2"             d1 -1 d2  1 d3 0;
ESTIMATE "T1 vs. T3"             d1 -1 d2  0 d3 1;
ESTIMATE "T2 vs. T3"             d1  0 d2 -1 d3 1;
RUN;
```

Intercepts are used only in predicted values.

Positive values indicate addition; negative values indicate subtraction.

Main Effects via Manual Contrasts: STATA

	Alt Group	Ref Group	Difference
Control vs. T1 =	$(\beta_0+\beta_1)$	$-\ (\beta_0)$	$= \beta_1$
Control vs. T2 =	$(\beta_0+\beta_2)$	$-\ (\beta_0)$	$= \beta_2$
Control vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0)$	$= \beta_3$
T1 vs. T2 =	$(\beta_0+\beta_2)$	$-\ (\beta_0+\beta_1)$	$= \beta_2 - \beta_1$
T1 vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0+\beta_1)$	$= \beta_3 - \beta_1$
T2 vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0+\beta_2)$	$= \beta_3 - \beta_2$

Note the order of the equations: the reference group mean *is subtracted from* the alternative group mean.

In SAS ESTIMATE statements (or SPSS TEST or STATA LINCOM), the variables refer to their fixed effects; the numbers refer to the operations of their fixed effects.

```
display "STATA Manual Contrasts for 4-Group Diffs"
regress y c.d1 c.d2 c.d3, level(95)
test (c.d1=0) (c.d2=0) (c.d3=0) // Omnibus F-test DF=3 group main effect
lincom _cons*1 + c.d1*0  + c.d2*0  + c.d3*0   // Control Mean
lincom _cons*1 + c.d1*1  + c.d2*0  + c.d3*0   // T1 Mean
lincom _cons*1 + c.d1*0  + c.d2*1  + c.d3*0   // T2 Mean
lincom _cons*1 + c.d1*0  + c.d2*0  + c.d3*1   // T3 Mean
lincom           c.d1*1  + c.d2*0  + c.d3*0   // Control vs T1
lincom           c.d1*0  + c.d2*1  + c.d3*0   // Control vs T2
lincom           c.d1*0  + c.d2*0  + c.d3*1   // Control vs T3
lincom           c.d1*-1 + c.d2*1  + c.d3*0   // T1 vs T2
lincom           c.d1*-1 + c.d2*0  + c.d3*1   // T1 vs T3
lincom           c.d1*0  + c.d2*-1 + c.d3*1   // T2 vs T3
```

Review of Single-Slope Interactions

- Previously we examined interactions involving either binary predictors or quantitative predictors with only linear slopes
  - **The role of a two-way interaction is to <u>adjust</u> its main effect slopes: to make them more/less positive or more/less negative**
  - However, the term "main effect slope" no longer applies: each becomes a ***simple slope*** that is ***conditional*** on each interacting predictor = 0
- e.g., $\boldsymbol{y_i} = \boldsymbol{\beta_0} + \boldsymbol{\beta_1}(\boldsymbol{w_i}) + \boldsymbol{\beta_2}(\boldsymbol{x_i}) + \boldsymbol{\beta_3}(\boldsymbol{z_i}) + \boldsymbol{\beta_4}(\boldsymbol{x_i})(\boldsymbol{z_i}) + \boldsymbol{e_i}$
  - w_i slope β_1 is a *marginal* main effect because it is not in an interaction
  - x_i slope β_2 is the *conditional* main effect of x_i *specifically when* $z_i = 0$
  - z_i slope β_3 is the *conditional* main effect of z_i *specifically when* $x_i = 0$
  - $x_i z_i$ slope β_4 is how β_2 differs per unit z_i, or how β_3 differs per unit x_i
- The $x_i z_i$ interaction here requires only one slope to test it when x_i and z_i are quantitative (or binary)—but if either predictor has 3+ categories, the $x_i z_i$ interaction would require more than 1 slope

Interactions with Manual Contrasts: SAS

- When using manual contrasts for predictors with 3 or more categories, **interactions must be specified with each separate contrast**
- For example, adding an interaction of 4-category group with age (0=85):

$$y_i = \beta_0 + \beta_1(d1_i) + \beta_2(d2_i) + \beta_3(d3_i) + \beta_4(Age_i - 85)$$
$$+\beta_5(d1_i)(Age_i - 85) + \beta_6(d2_i)(Age_i - 85) + \beta_7(d3_i)(Age_i - 85) + e_i$$

```
TITLE "SAS Group by Age for 4-Group Variable Included using Dummy Codes";
PROC GLM DATA=work.dataname NAMELEN=100;
MODEL y = d1 d2 d3 age d1*age d2*age d3*age / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
CONTRAST "Omnibus DF=3 SIMPLE effect F-test" d1 1, d2 1, d3 1;
CONTRAST "Omnibus DF=3 interaction F-test"   d1*age 1, d2*age 1, d3*age 1;

ESTIMATE "Age Slope for Control" age 1  d1*age  0 d2*age  0 d3*age  0;
ESTIMATE "Age Slope for T1"      age 1  d1*age  1 d2*age  0 d3*age  0;
ESTIMATE "Age Slope for T2"      age 1  d1*age  0 d2*age  1 d3*age  0;
ESTIMATE "Age Slope for T3"      age 1  d1*age  0 d2*age  0 d3*age  1;

ESTIMATE "Age Slope: Control vs. T1"    d1*age  1 d2*age  0 d3*age  0;
ESTIMATE "Age Slope: Control vs. T2"    d1*age  0 d2*age  1 d3*age  0;
ESTIMATE "Age Slope: Control vs. T3"    d1*age  0 d2*age  0 d3*age  1;
ESTIMATE "Age Slope: T1 vs. T2"         d1*age -1 d2*age  1 d3*age  0;
ESTIMATE "Age Slope: T1 vs. T3"         d1*age -1 d2*age  0 d3*age  1;
ESTIMATE "Age Slope: T2 vs. T3"         d1*age  0 d2*age -1 d3*age  1;
* Would also want to request simple group differences per age (or regions for them);
```

Interactions with Manual Contrasts: STATA

- When using manual contrasts for predictors with more than 2 categories, **interactions must be specified with each separate contrast**
- For example, adding an interaction of 4-category group with age (0=85):

$$y_i = \beta_0 + \beta_1(d1_i) + \beta_2(d2_i) + \beta_3(d3_i) + \beta_4(Age_i - 85) \\ + \beta_5(d1_i)(Age_i - 85) + \beta_6(d2_i)(Age_i - 85) + \beta_7(d3_i)(Age_i - 85) + e_i$$

```
display "STATA Group by Age for 4-Group Variable Included using Dummy Codes"
regress y c.d1 c.d2 c.d3 c.age c.d1#c.age c.d2#c.age c.d3#c.age, level(95)
test (c.d1=0) (c.d2=0) (c.d3=0) // Omnibus DF=3 SIMPLE effect F-test
test (c.d1#c.age=0) (c.d2#c.age=0) (c.d3#c.age=0) // DF=3 interaction F-test
lincom  c.age*1 + c.d1#c.age*0 + c.d2#c.age*0 + c.d3#c.age*0 // Age Slope for Control
lincom  c.age*1 + c.d1#c.age*1 + c.d2#c.age*0 + c.d3#c.age*0 // Age Slope for T1
lincom  c.age*1 + c.d1#c.age*0 + c.d2#c.age*1 + c.d3#c.age*0 // Age Slope for T2
lincom  c.age*1 + c.d1#c.age*0 + c.d2#c.age*0 + c.d3#c.age*1 // Age Slope for T3

lincom  c.d1#c.age*1  + c.d2#c.age*0  + c.d3#c.age*0 // Age Slope: Control vs T1
lincom  c.d1#c.age*0  + c.d2#c.age*1  + c.d3#c.age*0 // Age Slope: Control vs T2
lincom  c.d1#c.age*0  + c.d2#c.age*0  + c.d3#c.age*1 // Age Slope: Control vs T3
lincom  c.d1#c.age*-1 + c.d2#c.age*1  + c.d3#c.age*0 // Age Slope: T1 vs T2
lincom  c.d1#c.age*-1 + c.d2#c.age*0  + c.d3#c.age*1 // Age Slope: T1 vs T3
lincom  c.d1#c.age*0  + c.d2#c.age*-1 + c.d3#c.age*1 // Age Slope: T2 vs T3

// Would also want to request simple group differences per age (or regions for them)
```

Using BY/CLASS/i. statements instead

- Designate a predictor as "**categorical**" in program syntax
  - Put in on the CLASS statement in SAS; use i. prefix in STATA
- For a predictor with C categories, the program automatically then creates C new contrast variables, for example "group" with $C = 4$:

New Predictors Created Internally Mean this:	Control	Treat1	Treat2	Treat3
IsControl	1	0	0	0
IsTreat1	0	1	0	0
IsTreat2	0	0	1	0
IsTreat3	0	0	0	1

- It then figures out how many of these internal contrast variables are needed—if using an intercept (the default), then it's $C - 1$, not all C
- It enters them until it hits that criterion—if it leaves the last one out (as when you have an intercept), then last category becomes your reference
- Everywhere in syntax you refer to the categorical predictor, you must tell the program what to do with EACH of these internal contrast variables

Using BY/CLASS/i. statements instead

- Designate as "**categorical**" predictor in program syntax
  - If you let **SAS**/SPSS do the dummy coding via **CLASS**/BY, then the **highest/last group is default reference**
    - In SAS 9.4 you can change reference group: REF='level' | FIRST | LAST but it changes that group to be last in the data (→ confusing, so don't do it)
    - "Type III test of fixed effects" provide multivariate Wald tests by default
    - **LSMEANS**/EMMEANS can be used to get all cell means and comparisons without specifying each individual contrast, but you still have to ask for interaction contrasts (add / E to end of ESTIMATE to see the order of category values)
  - If you let STATA do the dummy coding via i.group, then the **lowest/first group is default reference**
    - Can change reference group, e.g., last = ref → ib(last).group
    - CONTRAST used to get omnibus tests (not provided by default)
    - MARGINS can be used to get all means and comparisons with much less code than describing each individual contrast
  - Btw, no such thing as "categorical" predictors in Mplus ☹
    - You must create contrasts manually for all grouping variables

Main Effects of Program-Categorical Predictors: SAS

```
TITLE "SAS Program-Created Contrasts for 4-Group Diffs via CLASS";
PROC GLM DATA=work.dataname NAMELEN=100;
CLASS group;
MODEL y = group / ALPHA=.05 CLPARM SS3
                  SOLUTION EFFECTSIZE;
LSMEANS group / DIFF=ALL;
```

CLASS statement means "make my contrast variables for me"

The LSMEANS line above gives you ALL of the following… note that one value has to be given for each possible level of the categorical predictor in *data* order

```
ESTIMATE "Control Mean" intercept 1 group 1 0 0 0;
ESTIMATE "T1 Mean"      intercept 1 group 0 1 0 0;
ESTIMATE "T2 Mean"      intercept 1 group 0 0 1 0;
ESTIMATE "T3 Mean"      intercept 1 group 0 0 0 1;
```

When predicting intercepts, 1 means "for that group only"

```
ESTIMATE "Control vs. T1"  group -1  1  0  0;
ESTIMATE "Control vs. T2"  group -1  0  1  0;
ESTIMATE "Control vs. T3"  group -1  0  0  1;
ESTIMATE "T1 vs. T2"       group  0 -1  1  0;
ESTIMATE "T1 vs. T3"       group  0 -1  0  1;
ESTIMATE "T2 vs. T3"       group  0  0 -1  1;
```

When predicting group differences, contrasts must sum to 0; here −1 = ref, 1 = alt, and 0 = ignore

```
CONTRAST "Omnibus DF=3 main effect F-test" group -1 1 0 0,
                                           group -1 0 1 0,
                                           group -1 0 0 1;
```

CLASS also gives this contrast by default

Can also make up whatever contrasts you feel like using DIVISOR option:

```
ESTIMATE "Mean of Treat groups" intercept 1 group  0 1 1 1 / DIVISOR=3;
ESTIMATE "Control vs. Mean of Treat groups" group -3 1 1 1 / DIVISOR=3;
RUN;
```

Main Effects of Program-Categorical Predictors: STATA

```
display "STATA Program-Created Contrasts for 4-Group Diffs"
display "i. means make my contrast variables for me (factor var)"
regress y ib(last).group, level(95)
contrast i.group // Omnibus DF=3 main effect F-test
margins i.group, pwcompare(pveffects) // Means per group and mean diffs
```

The MARGINS line above gives you ALL of the following… note that one value has to be given for each possible level of the categorical predictor in *data* order

```
lincom _cons*1 + 1.group*1  + 2.group*0  + 3.group*0  + 4.group*0 // Control Mean
lincom _cons*1 + 1.group*0  + 2.group*1  + 3.group*0  + 4.group*0 // T1 Mean
lincom _cons*1 + 1.group*0  + 2.group*0  + 3.group*1  + 4.group*0 // T2 Mean
lincom _cons*1 + 1.group*0  + 2.group*0  + 3.group*3  + 4.group*1 // T3 Mean
lincom           1.group*-1 + 2.group*1  + 3.group*0  + 4.group*0 // Control vs T1
lincom           1.group*-1 + 2.group*0  + 3.group*1  + 4.group*0 // Control vs T2
lincom           1.group*-1 + 2.group*0  + 3.group*0  + 4.group*1 // Control vs T3
lincom           1.group*0  + 2.group*-1 + 3.group*1  + 4.group*0 // T1 vs T2
lincom           1.group*0  + 2.group*-1 + 3.group*0  + 4.group*1 // T1 vs T3
lincom           1.group*0  + 2.group*0  + 3.group*-1 + 4.group*1 // T2 vs T3
```

Can also make up whatever contrasts you feel like (no DIVISOR option?) :

```
lincom _cons*1 + 1.group*0  + 2.group*.33 + 3.group*.33 + 4.group*.34 // Mean of Treat
lincom           1.group*-1 + 2.group*.33 + 3.group*.33 + 4.group*.34 // Cont v Treat
```

Interactions of Program-Categorical Predictors: SAS

For example, adding an interaction of group with age (0=85):

```
TITLE "SAS Group by Age for 4-Group Variable Included as Categorical";
PROC GLM DATA= work.dataname NAMELEN=100;
CLASS group;
MODEL y = group age group*age / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;

* To explain interaction as how group diffs depend on age:
LSMEANS group / DIFF=ALL AT (age)=(-5); * group intercept diffs at age 80;
LSMEANS group / DIFF=ALL AT (age)=(0);  * group intercept diffs at age 85;
LSMEANS group / DIFF=ALL AT (age)=(5);  * group intercept diffs at age 90;

* To explain interaction as how age slope depends on group:
ESTIMATE "Age Slope for Control" age 1  group*age 1 0 0 0;
ESTIMATE "Age Slope for T1"      age 1  group*age 0 1 0 0;
ESTIMATE "Age Slope for T2"      age 1  group*age 0 0 1 0;
ESTIMATE "Age Slope for T3"      age 1  group*age 0 0 0 1;

ESTIMATE "Age Slope: Control vs. T1"    group*age -1  1  0  0;
ESTIMATE "Age Slope: Control vs. T2"    group*age -1  0  1  0;
ESTIMATE "Age Slope: Control vs. T3"    group*age -1  0  0  1;
ESTIMATE "Age Slope: T1 vs. T2"         group*age  0 -1  1  0;
ESTIMATE "Age Slope: T1 vs. T3"         group*age  0 -1  0  1;
ESTIMATE "Age Slope: T2 vs. T3"         group*age  0  0 -1  1;
```

Can also make up whatever contrasts you feel like using DIVISOR option:

```
ESTIMATE "Mean Age Slope in Treat groups" age 1 group*age  0 1 1 1 / DIVISOR=3;
ESTIMATE "Age Slope: Control vs. Mean of Treat" group*age -3 1 1 1 / DIVISOR=3;
RUN;
```

Interactions of Program-Categorical Predictors: STATA

For example, adding an interaction of group with age (0=85):

```
display "STATA Group by Age for 4-Group Variable Included as Categorical"
regress y ib(last).group c.age ib(last).group#c.age, level(95)
contrast i.group       // Omnibus DF=3 simple effect F-test
contrast i.group#c.age // DF=3 interaction F-test
lincom  c.age*1 + i1.group#c.age*1 // Age Slope for Cont
lincom  c.age*1 + i2.group#c.age*1 // Age Slope for T1
lincom  c.age*1 + i3.group#c.age*1 // Age Slope for T2
lincom  c.age*1 + i4.group#c.age*1 // Age Slope for T3

lincom  i1.group#c.age*-1 + i2.group#c.age*1 // Age Slope: Cont vs T1
lincom  i1.group#c.age*-1 + i3.group#c.age*1 // Age Slope: Cont vs T2
lincom  i1.group#c.age*-1 + i4.group#c.age*1 // Age Slope: Cont vs T3
lincom  i2.group#c.age*-1 + i3.group#c.age*1 // Age Slope: T1 vs T2
lincom  i2.group#c.age*-1 + i4.group#c.age*1 // Age Slope: T1 vs T3
lincom  i3.group#c.age*-1 + i4.group#c.age*1 // Age Slope: T2 vs T3
```

Can also make up whatever contrasts you feel like (no DIVISOR option?) :

```
lincom c.age*1 i1.group#c.age*0   + i2.group#c.age*.33 /// Age Slope for Treat
               i1.group#c.age*.33 + i2.group#c.age*.34

lincom         i1.group#c.age*-1  + i2.group#c.age*.33 /// Age Slope: C vs Treat
               i1.group#c.age*.33 + i2.group#c.age*.34
```

Program-Categorical Predictors → Marginal Effects

- Letting the program build contrasts for categorical predictors (instead of creating manual dummy codes) does the following:
  - Allows LSMEANS/EMMEANS/MARGINS (for cell means and differences)
  - Provides omnibus (multiple DF) multivariate Wald tests for group effects
  - **Marginalizes the group effect across interacting predictors** → omnibus F-tests represent marginal main effects (instead of simple)
  - ```
    MODEL    y = group sexMW group*sexMW
    regress y ib(last).group sexMW ib(last).group#sexMW,
    ```
    (in which group is always "categorical")

Type 3 Tests of Fixed Effects	Interpretation if sexMW is "quantitative" (no CLASS/i)	Interpretation if sexMW is "categorical" on CLASS/i
sexMW	Marginal diff across groups	Marginal diff across groups
group	Group diff if sexMW=0	Marginal diff across sexes
group*sexMW	Interaction	Interaction
    ```

Interactions Among Program-**Categorical** Predictors: Default ANOVA Output

- Traditional ANOVA model includes **all possible higher-order** interactions among categorical predictors… by default!
  - Software does this for you; nonsignificant interactions usually still are kept in the model (but only significant interactions are interpreted)
  - This is very different from typical practice in "multiple regression"!
- Omnibus **marginal** main effects are provided by default
  - i.e., what we ask for via CONTRAST using manual group contrasts
  - But are **basically useless** given significant interactions
- Omnibus **interaction effects** are provided
  - i.e., what we ask for via CONTRAST using manual group contrasts
  - But **need to be split into DF=1 effects** to understand the interaction

- In Example 7 we'll see how to make software give us more useful info… simple main effects and specific interaction contrasts to the rescue!

Multiple-DF Interactions More Generally

- Interactions can be tested between any predictors, including quantitative predictors that require more than one slope…
- Do piecewise education slopes differ between men and women? *(inspired by Example 4 models predicting annual income)*

$$\boldsymbol{Income_i = \beta_0 + \beta_1(lessHS_i) + \beta_2(gradHS_i) + \beta_3(overHS_i) + \beta_4(MvW_i)}$$
$$\boldsymbol{+\beta_5(MvW_i)(lessHS_i) + \beta_6(MvW_i)(gradHS_i) + \beta_7(MvW_i)(overHS_i) + e_i}$$

  - Use SAS CONTRAST or STATA TEST to lump together $\boldsymbol{\beta_5}$, $\boldsymbol{\beta_6}$, and $\boldsymbol{\beta_7}$ for DF=3 F-test of interaction term
  - Simple slopes $\boldsymbol{\beta_1}$, $\boldsymbol{\beta_2}$, and $\boldsymbol{\beta_3}$ give education effect for $MvW_i = 0$
  - Interactions $\boldsymbol{\beta_5}$, $\boldsymbol{\beta_6}$, and $\boldsymbol{\beta_7}$ give DIFF in education effect for $MvW_i = 1$
  - So simple slopes for each subsample of education for $MvW_i = 1$ are given by: $\boldsymbol{\beta_1}+\boldsymbol{\beta_5}$ for $lessHS_i$, $\boldsymbol{\beta_2}+\boldsymbol{\beta_6}$ for $gradHS_i$, and $\boldsymbol{\beta_3}+\boldsymbol{\beta_7}$ for $overHS_i$

- Btw, adding a third sex category (e.g., nonbinary) would require one more simple main effect for it, as well as three more interaction terms!

Categorical Predictors with Issues

- Experimental designs with fully crossed conditions lend themselves to analysis of variance-type models
- What happens when things go wrong? Two examples:
  - ANOVA with a hole in it
  - Predictors that don't apply or weren't measured for everyone
- These designs can be analyzed using **nested effects**
  - Different programs specify these differently, so I'll show them using a common language of pseudo-interaction terms
  - In specifying nested effects, what look like "interactions" actually act as switches instead to turn effects on/off…

A Traditional View of ANOVA

ANOVAs usually provide F-tests for **marginal mean differences…**

Is this **really** what you want to know?

F(df=1) → a $v.$ b

F(df=2) → 1 $v.$ 2. $v.$ 3

F(df=2) → a $v.$ b * 1 $v.$ 2. $v.$ 3

Means	$\overline{1}$	$\overline{2}$	$\overline{3}$
$\bar{a}$	$\overline{a1}$	$\overline{a2}$	$\overline{a3}$
$\bar{b}$	$\overline{b1}$	$\overline{b2}$	$\overline{b3}$

ANOVA as a General Linear Model

$$
\begin{aligned}
y_i = &\boxed{\beta_0} + \boxed{\beta_1(a1\ v.b1_i)} \\
&+ \boxed{\beta_2(a1\ v.a2_i) + \beta_3(a1\ v.a3_i)} \\
&+ \beta_4(a1\ v.b1_i)(a1\ v.a2_i) \\
&+ \beta_5(a1\ v.b1_i)(a1\ v.a3_i) + e_i
\end{aligned}
$$

The focus is now on **differences between *specific* conditions** as created by the β fixed effects.

Means	$\overline{1}$	$\overline{2}$	$\overline{3}$
$\bar{a}$	$\overline{a1}$	$\overline{a2}$	$\overline{a3}$
$\bar{b}$	$\overline{b1}$	$\overline{b2}$	$\overline{b3}$

ANOVA as a General Linear Model

- Software will find any **simple slopes (differences) you ask for**
  - TEST in SPSS MIXED (not GLM); ESTIMATE in SAS (GLM or MIXED)
  - LINCOM or MARGINS in STATA; NEW in Mplus

- Seeing research questions through linear models saves **nontraditional research designs**
  - Not fully crossed on purpose or by accident… "ANOVA with a hole in it"

Means	$\bar{1}$	$\bar{2}$	$\bar{3}$
$\bar{a}$	β_0	$\beta_0 + \beta_2$	$\beta_0 + \beta_3$
$\bar{b}$	$\beta_0 + \beta_1$	$\beta_0 + \beta_1 + \beta_2 + \boxed{\beta_4}$	$\beta_0 + \beta_1 + \beta_3 + \boxed{\beta_5}$

A Nontraditional ANOVA Design

$$y_i = \boxed{\beta_0} + \boxed{\beta_1(t3\ v.t1_i) + \ \beta_2(t2\ v.t1_i)}$$
$$+\boxed{\beta_3(t1_i)(t\ v.c_i)} + \boxed{\beta_4(t2_i)(t\ v.c_i)} + e_i$$

β_3 and β_4 are not interaction terms. Instead, they are *nested* effects.

You are allowed to use any C effects you want to represent the C means, even in fully crossed designs!

Means	Cohort 1	Cohort 2	Cohort 3
Control	$\beta_0 + \beta_1 + \boxed{\beta_3}$	$\beta_0 + \beta_2 + \boxed{\beta_4}$	
Treatment	$\beta_0 + \boxed{\beta_1}$	$\beta_0 + \boxed{\beta_2}$	$\boxed{\beta_0}$

A Nested-Effects General Linear Model

- Example: predicting outcomes by dementia type and dementia timing in persons with OR without dementia
  - Type and timing do not apply to persons without dementia
  - So this requires the following new variables…

```
* Create a switch variable and nested type variable;
IF demtype="none" THEN DO; demYes=0; demAorV=  0; END;
IF demtype="AD"   THEN DO; demYes=1; demAorV=-.5; END;
IF demtype="VA"   THEN DO; demYes=1; demAorV= .5; END;
* Create a timing variable (0=5 years) when applicable;
IF demtype="none" THEN DO; demtime5=0;          END;
IF demtype="AD"   THEN DO; demtime5=demtime-5; END;
IF demtype="VA"   THEN DO; demtime5=demtime-5; END;
```

A Nested-Effects General Linear Model

$$y_i = \beta_0 + \beta_1(demYes_i) + \beta_2(demAorV_i) \\ +\beta_3(demYes_i)(demtime_i - 5) + \beta_4(demAorV_i)(demtime_i - 5) + e_i$$

Fixed Effect	**Interpretation**
β_0: Intercept	Expected outcome for persons without dementia
β_1: demYes	Simple slope for difference between persons without dementia or with dementia at 5 years (averaged across AD and VA dementia types)
β_2: demAorV	Simple slope for difference between persons with AD or VA type dementia (at 5 years)
β_3: demYes* demtime5	Is NOT an interaction term: Slope for difference in outcome per year of dementia *only in persons with dementia* (averaged across AD and VA dementia types)
β_4: demAorV* demtime5	IS an interaction term: Difference in slope for effect of years between persons with AD or VA type

Other Uses for GLM Nested Effects

- **Nested effects** are main effects specified to apply selectively to subsamples of the possible cases contributing to the model
- They have lots of potential—but relatively unknown—uses
  - "If and how much" effects of **semi-continuous** predictors
    - Difference between groups of "younger" and "older" adults; + slope for years of age within "older" adults (see Hoffman 2015 ch. 12)
    - Presence and severity of abuse: difference between groups of "not abused" and "abused" persons; + slope for severity of abuse within "abused" group (for which severity > 0)
  - Missing, refused to answer, or other **incomplete predictor data**:
    - Difference between groups of "incomplete" versus "complete" predictor values; + slope for predictor values in "complete" group
  - Predictor effects that only apply to one outcome in a multivariate GLM predicting multiple outcomes simultaneously…
    - Come back to my *Generalized Linear Models* class to see this usage!